Chapter 19
Classical Nonlinear Oscillator

19.1 Statement of the Problem

Let's consider one-dimensional motion of a particle with mass m near a minimum of an arbitrary smooth potential [19]

$$U(x) = \frac{kx^2}{2} + V(x), \qquad V(x) = O\left(x^3\right)$$

(we have chosen the origin of x and the zero energy level to be at the minimum). If we neglect $V(x)$, then the equation of motion

$$m\frac{\mathrm{d}^2x}{\mathrm{d}t^2} = -\frac{\mathrm{d}U}{\mathrm{d}x}$$

becomes

$$\frac{\mathrm{d}^2x}{\mathrm{d}t^2} + \omega_0^2 x = 0, \qquad \omega_0^2 = \frac{k}{m},$$

and has the solution

$$x(t) = a\cos\omega_0 t + b\sin\omega_0 t.$$

Now we consider the effect of

$$V(x) = \sum_{n=1}^{\infty} c_n x^{n+2}.$$

Choosing units of measurement in such a way that $m = 1$ and $k = 1$, we have the equation of motion

$$\frac{\mathrm{d}^2x}{\mathrm{d}t^2} + x = R(x) \equiv -\frac{\mathrm{d}V}{\mathrm{d}x}.$$

Its solution $x(t)$ is a periodic function of t. If we choose the time origin at a maximum of $x(t)$, then $x(t)$ is an even function, due to reversibility. In the zeroth approximation $x(t) = a\cos t$.

A. Grozin, *Introduction to Mathematica® for Physicists*, Graduate Texts in Physics, 145
DOI 10.1007/978-3-319-00894-3_19, © Springer International Publishing Switzerland 2014

In[1] := V = Series[c[1] ∗ x^3, {x, 0, 3}]
Out[1] $= c[1]x^3 + O[x]^4$
In[2] := R = −D[V, x]
Out[2] $= -3c[1]x^2 + O[x]^3$
In[3] := x[t] = Series[a ∗ Cos[t], {a, 0, 1}]
Out[3] $= \mathrm{Cos}[t]a + O[a]^2$
The equation of motion is satisfied at $O(a)$.
In[4] := D[x[t], {t, 2}] + x[t]
Out[4] $= O[a]^2$

19.2 The First Correction

Now we want to take terms of order a^2 into account. The right-hand side is
In[5] := R1 = R/.x−>x[t]
Out[5] $= -3\left(c[1]\mathrm{Cos}[t]^2\right)a^2 + O[a]^3$
Let's expand it in harmonics.
In[6] := R1 = Map[TrigReduce, R1]
Out[6] $= -\frac{3}{2}(c[1] + c[1]\mathrm{Cos}[2t])a^2 + O[a]^3$
That is, the "driving force" contains zeroth and second harmonics. This means that we should add such harmonics to $x(t)$. We'll not add the solution of the homogeneous equation—the first harmonic: by definition of the amplitude a, it is completely given by the leading term $a\cos t$.
In[7] := x[t] = Series[a ∗ Cos[t]+
a^2 ∗ Sum[b[2, j] ∗ Cos[j ∗ t], {j, 0, 2, 2}], {a, 0, 2}]
Out[7] $= \mathrm{Cos}[t]a + (b[2,0] + b[2,2]\mathrm{Cos}[2t])a^2 + O[a]^3$
Now we substitute this form of the solution into the equation of motion.
In[8] := Eq = D[x[t], {t, 2}] + x[t] − (R/.x−>x[t])
Out[8] $= \left(b[2,0] + 3c[1]\mathrm{Cos}[t]^2 - 3b[2,2]\mathrm{Cos}[2t]\right)a^2 + O[a]^3$
In[9] := Eq = Map[TrigReduce, Eq]
Out[9] $= \frac{1}{2}(2b[2,0] + 3c[1] - 6b[2,2]\mathrm{Cos}[2t] + 3c[1]\mathrm{Cos}[2t])a^2 + O[a]^3$
In[10] := Eq2 = SeriesCoefficient[Eq, 2]
Out[10] $= \frac{1}{2}(2b[2,0] + 3c[1] - 6b[2,2]\mathrm{Cos}[2t] + 3c[1]\mathrm{Cos}[2t])$
This expression should vanish. How can we separate harmonics? Let's help *Mathematica* a little.
In[11] := Eq2 = Eq2/.Cos[j_. ∗ t]−>z^j
Out[11] $= \frac{1}{2}\left(2b[2,0] - 6z^2b[2,2] + 3c[1] + 3z^2c[1]\right)$
The coefficients of z^0 and z^2 should vanish.
In[12] := Eq20 = Coefficient[Eq2, z, 0]
Out[12] $= \frac{1}{2}(2b[2,0] + 3c[1])$

In[13] := Eq22 = Coefficient[Eq2, z, 2]

Out[13] $= \frac{1}{2}(-6b[2,2] + 3c[1])$

We can find $b[2,0]$ from the first equation and $b[2,2]$ from the second one.

In[14] := b[2,0] = b[2,0]/. Solve[Eq20 == 0, b[2,0]][[1]]

Out[14] $= -\frac{3c[1]}{2}$

In[15] := b[2,2] = b[2,2]/. Solve[Eq22 == 0, b[2,2]][[1]]

Out[15] $= \frac{c[1]}{2}$

Now we know the solution.

In[16] := x[t] = x[t]

Out[16] $= \mathrm{Cos}[t]a + \left(-\frac{3c[1]}{2} + \frac{1}{2}c[1]\mathrm{Cos}[2t]\right)a^2 + O[a]^3$

Let's check energy conservation.

In[17] := Et = D[x[t], t]^2/2 + x[t]^2/2 + (V/.x−>x[t]);

In[18] := Map[TrigReduce, Et]

Out[18] $= \frac{a^2}{2} + O[a]^4$

In[19] := Clear[b]

19.3 The Second Correction

Now we want to find two corrections.

In[20] := n = 2;

In[21] := V = Series[Sum[c[i] ∗ x^(i + 2), {i, 1, n}], {x, 0, n + 2}]

Out[21] $= c[1]x^3 + c[2]x^4 + O[x]^5$

In[22] := R = −D[V, x]

Out[22] $= -3c[1]x^2 - 4c[2]x^3 + O[x]^4$

This is $x[t]$ up to a^2.

In[23] := x[t] = Series[a ∗ Cos[t]+
a^2 ∗ Sum[b[2, j] ∗ Cos[j ∗ t], {j, 0, 2, 2}], {a, 0, n}]

Out[23] $= \mathrm{Cos}[t]a + (b[2,0] + b[2,2]\mathrm{Cos}[2t])a^2 + O[a]^3$

The right-hand side of the equation of motion.

In[24] := R1 = Map[TrigReduce, ExpandAll[R/.x−>x[t]]]

Out[24] $= -\frac{3}{2}(c[1] + c[1]\mathrm{Cos}[2t])a^2 +$
$(-6b[2,0]c[1]\mathrm{Cos}[t] - 3b[2,2]c[1]\mathrm{Cos}[t] - 3c[2]\mathrm{Cos}[t] -$
$3b[2,2]c[1]\mathrm{Cos}[3t] - c[2]\mathrm{Cos}[3t])a^3 +$
$O[a]^4$

There are the first and the third harmonics at the order a^3, that is, there is a resonant term in the "driving force" which would lead to an unbounded growth of the solution. This means we have forgotten something. Namely, we have forgotten that the oscillation period depends on the amplitude (unless the potential is strictly parabolic). And our solution should contain $\cos(j\omega t)$. If we denote $\tau = \omega t$, then the equation of motion is

$$\omega^2 \frac{d^2x}{d\tau^2} + x = R\,.$$

Let's suppose that the variable t in the program really means τ and denote $\omega^2 = w$. It is a series in a^2 beginning with 1.

In[25] := w = Series[1 + Sum[u[i] * a^i, {i, 2, n + 1, 2}], {a, 0, n + 1}]

Out[25] $= 1 + u[2]a^2 + O[a]^4$

Now we are able to cancel the first harmonic in the a^3 term of the equation of motion. And the third one should be added to the general form of the solution.

**In[26] := x[t] = Series[a * Cos[t] + a^2 * Sum[b[2, j] * Cos[j * t], {j, 0, 2, 2}]+
a^3 * Sum[b[3, j] * Cos[j * t], {j, 3, 3, 2}], {a, 0, n + 1}]**

Out[26] $= \mathrm{Cos}[t]a + (b[2,0] + b[2,2]\mathrm{Cos}[2t])a^2 + b[3,3]\mathrm{Cos}[3t]a^3 + O[a]^4$

The equation of motion is

**In[27] := Eq = Map[TrigReduce,
ExpandAll[w * D[x[t], {t, 2}] + x[t] − (R/.x−>x[t])]]**

Out[27] $= \frac{1}{2}(2b[2,0] + 3c[1] - 6b[2,2]\mathrm{Cos}[2t] + 3c[1]\mathrm{Cos}[2t])a^2 +$
$(6b[2,0]c[1]\mathrm{Cos}[t] + 3b[2,2]c[1]\mathrm{Cos}[t] + 3c[2]\mathrm{Cos}[t] -$
$8b[3,3]\mathrm{Cos}[3t] + 3b[2,2]c[1]\mathrm{Cos}[3t] + c[2]\mathrm{Cos}[3t] - \mathrm{Cos}[t]u[2])a^3 +$
$O[a]^4$

We already know how to solve it at the order a^2.

In[28] := Eq2 = SeriesCoefficient[Eq, 2]/.Cos[j_. * t]−>z^j

Out[28] $= \frac{1}{2}\left(2b[2,0] - 6z^2b[2,2] + 3c[1] + 3z^2c[1]\right)$

**In[29] := Do[Print[b[2, j] = b[2, j]/.Solve[Coefficient[Eq2, z, j] == 0, b[2, j]][[1]]],
{j, 0, 2, 2}]**

$-\frac{3c[1]}{2}$

$\frac{c[1]}{2}$

At the order a^3:

In[30] := Eq3 = SeriesCoefficient[Eq, 3]/.Cos[j_. * t]−>z^j

Out[30] $= -8z^3b[3,3] - \frac{15}{2}zc[1]^2 + \frac{3}{2}z^3c[1]^2 + 3zc[2] + z^3c[2] - zu[2]$

This is the coefficient of the first harmonic, i.e., of z^1:

In[31] := Eq31 = Coefficient[Eq3, z, 1]

Out[31] $= -\frac{15}{2}c[1]^2 + 3c[2] - u[2]$

It can be nullified by choosing $u[2]$.

In[32] := u[2] = u[2]/.Solve[Eq31 == 0, u[2]][[1]]

Out[32] $= -\frac{3}{2}\left(5c[1]^2 - 2c[2]\right)$

And this is the coefficient of the third harmonic, i.e., of z^3:

In[33] := Eq33 = Coefficient[Eq3, *z*, 3]

$$\mathrm{Out}[33] = -8b[3,3] + \frac{3c[1]^2}{2} + c[2]$$

It can be nullified by choosing $b[3,3]$.

In[34] := *b*[3,3] = *b*[3,3]/.Solve[Eq33 == 0, *b*[3,3]][[1]]

$$\mathrm{Out}[34] = \frac{1}{16}\left(3c[1]^2 + 2c[2]\right)$$

Now we know the oscillation frequency

In[35] := *w* = *w*

$$\mathrm{Out}[35] = 1 - \frac{3}{2}\left(5c[1]^2 - 2c[2]\right)a^2 + O[a]^4$$

and $x[t]$:

In[36] := *x*[*t*] = *x*[*t*]

$$\mathrm{Out}[36] = \mathrm{Cos}[t]a + \left(-\frac{3c[1]}{2} + \frac{1}{2}c[1]\mathrm{Cos}[2t]\right)a^2 + \frac{1}{16}\left(3c[1]^2 + 2c[2]\right)\mathrm{Cos}[3t]a^3 + O[a]^4$$

Let's check energy conservation.

**In[37] := Et = Map[TrigReduce,
ExpandAll[*w* ∗ *D*[*x*[*t*], *t*]^2/2 + *x*[*t*]^2/2 + (*V*/.*x*−>*x*[*t*])]]**

$$\mathrm{Out}[37] = \frac{a^2}{2} + \frac{1}{16}\left(-37c[1]^2 + 18c[2]\right)a^4 + O[a]^5$$

In[38] := Clear[*b*, *u*]

19.4 The *n*th Correction

Now we'll write a program which can find n corrections in a to the particle motion for any n. Just a single line should be changed for the calculation with a new value of n.

In[39] := *n* = 4;

The correction to the potential and the "driving force."

In[40] := *V* = Series[Sum[*c*[*i*] ∗ *x*^(*i* + 2), {*i*, 1, *n*}], {*x*, 0, *n* + 2}]

$$\mathrm{Out}[40] = c[1]x^3 + c[2]x^4 + c[3]x^5 + c[4]x^6 + O[x]^7$$

In[41] := *R* = −*D*[*V*, *x*]

$$\mathrm{Out}[41] = -3c[1]x^2 - 4c[2]x^3 - 5c[3]x^4 - 6c[4]x^5 + O[x]^6$$

The frequency squared is a series in a^2.

In[42] := *w* = Series[1 + Sum[*u*[*i*] ∗ *a*^*i*, {*i*, 2, *n* + 1, 2}], {*a*, 0, *n* + 1}]

$$\mathrm{Out}[42] = 1 + u[2]a^2 + u[4]a^4 + O[a]^6$$

The general form of the solution. The order a^i contains harmonics up to the ith one. They are all even at even values of i and odd at odd values. The first harmonic never appears—by definition, it is entirely contained in the leading term $a\cos t$.

```
In[43] := x[t] = Series[a*Cos[t]+
        Sum[a^i*Sum[b[i,j]*Cos[j*t],{j,If[EvenQ[i],0,3],i,2}],{i,2,n+1}],
    {a,0,n+1}]
```

$$\mathrm{Out}[43] = \mathrm{Cos}[t]a + (b[2,0] + b[2,2]\mathrm{Cos}[2t])a^2 + b[3,3]\mathrm{Cos}[3t]a^3 + (b[4,0] + b[4,2]\mathrm{Cos}[2t] + b[4,4]\mathrm{Cos}[4t])a^4 + (b[5,3]\mathrm{Cos}[3t] + b[5,5]\mathrm{Cos}[5t])a^5 + O[a]^6$$

The equation of motion.

```
In[44] := Eq = Map[TrigReduce,
    ExpandAll[w*D[x[t],{t,2}] + x[t] - (R/.x->x[t])]]
```

$$\mathrm{Out}[44] = \frac{1}{2}(2b[2,0] + 3c[1] - 6b[2,2]\mathrm{Cos}[2t] + 3c[1]\mathrm{Cos}[2t])a^2 + (6b[2,0]c[1]\mathrm{Cos}[t] + 3b[2,2]c[1]\mathrm{Cos}[t] + 3c[2]\mathrm{Cos}[t] - 8b[3,3]\mathrm{Cos}[3t] + 3b[2,2]c[1]\mathrm{Cos}[3t] + c[2]\mathrm{Cos}[3t] - \mathrm{Cos}[t]u[2])a^3 + \frac{1}{8}\left(8b[4,0] + 24b[2,0]^2c[1] + 12b[2,2]^2c[1] + 48b[2,0]c[2] + 24b[2,2]c[2] + 15c[3] - 24b[4,2]\mathrm{Cos}[2t] + 48b[2,0]b[2,2]c[1]\mathrm{Cos}[2t] + 24b[3,3]c[1]\mathrm{Cos}[2t] + 48b[2,0]c[2]\mathrm{Cos}[2t] + 48b[2,2]c[2]\mathrm{Cos}[2t] + 20c[3]\mathrm{Cos}[2t] - 120b[4,4]\mathrm{Cos}[4t] + 12b[2,2]^2c[1]\mathrm{Cos}[4t] + 24b[3,3]c[1]\mathrm{Cos}[4t] + 24b[2,2]c[2]\mathrm{Cos}[4t] + 5c[3]\mathrm{Cos}[4t] - 32b[2,2]\mathrm{Cos}[2t]u[2]\right)a^4 + \left(3b[2,2]b[3,3]c[1]\mathrm{Cos}[t] + 6b[4,0]c[1]\mathrm{Cos}[t] + 3b[4,2]c[1]\mathrm{Cos}[t] + 12b[2,0]^2c[2]\mathrm{Cos}[t] + 12b[2,0]b[2,2]c[2]\mathrm{Cos}[t] + 6b[2,2]^2c[2]\mathrm{Cos}[t] + 3b[3,3]c[2]\mathrm{Cos}[t] + 15b[2,0]c[3]\mathrm{Cos}[t] + 10b[2,2]c[3]\mathrm{Cos}[t] + \frac{15}{4}c[4]\mathrm{Cos}[t] - 8b[5,3]\mathrm{Cos}[3t] + 6b[2,0]b[3,3]c[1]\mathrm{Cos}[3t] + 3b[4,2]c[1]\mathrm{Cos}[3t] + 3b[4,4]c[1]\mathrm{Cos}[3t] + 12b[2,0]b[2,2]c[2]\mathrm{Cos}[3t] + 3b[2,2]^2c[2]\mathrm{Cos}[3t] + 6b[3,3]c[2]\mathrm{Cos}[3t] + 5b[2,0]c[3]\mathrm{Cos}[3t] + \frac{15}{2}b[2,2]c[3]\mathrm{Cos}[3t] + \frac{15}{8}c[4]\mathrm{Cos}[3t] - 24b[5,5]\mathrm{Cos}[5t] + 3b[2,2]b[3,3]c[1]\mathrm{Cos}[5t] + 3b[4,4]c[1]\mathrm{Cos}[5t] + 3b[2,2]^2c[2]\mathrm{Cos}[5t] + 3b[3,3]c[2]\mathrm{Cos}[5t] + \frac{5}{2}b[2,2]c[3]\mathrm{Cos}[5t] + \frac{3}{8}c[4]\mathrm{Cos}[5t] - 9b[3,3]\mathrm{Cos}[3t]u[2] - \mathrm{Cos}[t]u[4]\right)a^5 + O[a]^6$$

All terms of the orders a^i for i from 2 to $n+1$ must vanish. If i is odd, the first harmonic is present; a correction to the frequency squared is found from the condition that this harmonic vanishes. All other harmonics give us coefficients in $x(t)$.

```
In[45] := Do[Eqi = SeriesCoefficient[Eq,i]/.Cos[j_.*t]->z^j;
        If[OddQ[i],
            u[i-1] = u[i-1]/.Solve[Coefficient[Eqi,z,1] == 0,u[i-1]][[1]]];
        Do[b[i,j] = b[i,j]/.Solve[Coefficient[Eqi,z,j] == 0,b[i,j]][[1]],
            {j,If[EvenQ[i],0,3],i,2}],
    {i,2,n+1}]
```

Now we know the frequency squared.

In[46] := w = w

$$\text{Out[46]} = 1 - \frac{3}{2}\left(5c[1]^2 - 2c[2]\right)a^2 - \frac{3}{32}\left(335c[1]^4 - 572c[1]^2c[2] - 4c[2]^2 + 280c[1]c[3] - 40c[4]\right)a^4 + O[a]^6$$

and $x(t)$

In[47] := x[t] = x[t]

$$\text{Out[47]} = \text{Cos}[t]a + \left(-\frac{3c[1]}{2} + \frac{1}{2}c[1]\text{Cos}[2t]\right)a^2 + \frac{1}{16}\left(3c[1]^2 + 2c[2]\right)\text{Cos}[3t]a^3 + \left(-\frac{3}{8}\left(19c[1]^3 - 20c[1]c[2] + 5c[3]\right) + \frac{1}{48}\left(177c[1]^3 - 186c[1]c[2] + 40c[3]\right)\text{Cos}[2t] + \frac{1}{48}\left(3c[1]^3 + 6c[1]c[2] + 2c[3]\right)\text{Cos}[4t]\right)a^4 + \left(\frac{3}{256}\left(237c[1]^4 - 172c[1]^2c[2] - 28c[2]^2 - 12c[1]c[3] + 20c[4]\right)\text{Cos}[3t] + \frac{1}{768}\left(15c[1]^4 + 60c[1]^2c[2] + 12c[2]^2 + 44c[1]c[3] + 12c[4]\right)\text{Cos}[5t]\right)a^5 + O[a]^6$$

Let's check energy conservation.

**In[48] := Et = Map[TrigReduce,
ExpandAll[w ∗ D[x[t], t]^2/2 + x[t]^2/2 + (V/.x−>x[t])]]**

$$\text{Out[48]} = \frac{a^2}{2} + \frac{1}{16}\left(-37c[1]^2 + 18c[2]\right)a^4 + \frac{1}{1536}\left(-9309c[1]^4 + 17796c[1]^2c[2] + 300c[2]^2 - 10880c[1]c[3] + 1920c[4]\right)a^6 + O[a]^7$$

It is easy to write a function with the parameter *n* which can be used as a black box. It should use only local variables.

Now we save the results for the energy Et and the frequency squared *w* to a file; later we'll compare them to the similar results in quantum mechanics.

**In[49] := Ec = Normal[Et]/.a−>Sqrt[2 ∗ A];
Wc = Normal[Simplify[Sqrt[w]]]/.a−>Sqrt[2 ∗ A];
Save["class.m", {Ec, Wc}]**